

KASINO

EN SIMULATOR.

VED BØRGE CHRISTENSEN, DATO, TØNDER.

Ved datatræffestævnet i Tønder i marts måned 1977 var der et oplæg til et projektarbejde: "Spillekasino", som deltagerne kunne arbejde på i grupper.

Som en service overfor de deltagere, som ikke nåede at blive færdige, og overfor de medlemmer, der ikke kunne deltage i mødet, har redaktionen formået Børge Christensen til at skrive hele simulatoren med kommentarer og programudskrift.

En sådan service vil vi yde overfor medlemmerne, når lignende projekter kommer redaktionen i hænde.

KASINO: system

består af:

en CROUPIER

datastruktur: står ved HJULEt overfor SPILLERen.

Kan tale med SPILLERen og starte HJULEt.

handlingsmønster: Beder SPILLERen gøre sin indsats og starter derpå HJULEt. Dette gør CROUPIERen for hvert nyt spil.slut CROUPIER

et HJUL

datastruktur: HJULEt er ca. 1 m i diameter og er "lagkagedelt" i 9 lige store afsnit. Tre afsnit er mærket med cifret 1, tre afsnit er mærket med cifret 2, to afsnit er mærket med cifret 3, og ét afsnit er mærket med cifret 4.handlingsmønster: HJULEt startes af CROUPIERen og standser af sig selv efter nogle få omløb. Når det er standset, står ét af de ni afsnit ud for en fast monteret viser. Det tal, som står noteret på det udpegede afsnit, er spillets resultat.slut HJUL

en SPILLER

datastruktur: SPILLERen medbringer et vist beløb, som han har lyst til at sætte spil i håb om gevinst. SPILLERen kan se HJULEt, samt se og høre CROUPIERen og BANKØRen.handlingsmønster: SPILLERen kan betale penge til BANKØRen, som opretter en konto for SPILLERen. Når CROUPIERen beder SPILLERen gøre sin indsats, kan han enten vælge ét af tallene 1, 2, 3 eller 4, eller opgive at spille videre. Hvis han ønsker at fortsætte, skal han efter at have valgt sit nummer angive sin indsats (kun hele kroner) til BANKØRen. Hvis SPILLERen ønsker at forlade spillet, får han af BANKØRen udbetalt det beløb, der står på kontoen.slut SPILLER

en BANKØR

datastruktur: holder under hele spillet SPILLERens konto. Kan se HJULEt og dermed det enkelte spils udfald. Kender for hvert enkelt spil det nummer, SPILLERen har valgt, og det beløb, der er sat på spil. Kan betjene KLOKKEn.

handlingsmønster: modtager penge af SPILLERen ved spillets begyndelse eller når SPILLERen under spillet har tømt sin konto og ønsker at fortsætte. BANKØREN registrerer før hvert spil SPILLERens indsats. Såfremt denne indsats er højere end det beløb, der står på SPILLERens konto, giver BANKØREN SPILLERen valget mellem at indbetale flere penge på sin konto eller gøre en mindre indsats. Hvis SPILLERen ikke vil nogen af delene, beder BANKØREN ham forlade spillet. Hvis SPILLERen forsøger at sætte et ikke helt antal kroner på spil, beder BANKØREN ham ændre indsatsen til et helt antal kroner.

Når HJULEt er standset og udfaldet af et spil foreligger, vil BANKØREN i tilfælde af tab trække det beløb, der er sat på spil, fra SPILLERens konto, og i tilfælde af gevinst regner BANKØREN gevinsten og lægger den til SPILLERens konto.

Gevinsten udregnes af BANKØREN således:

Hvis der vindes på nummer 1 eller 2, er gevinsten det dobbelte af indsatsen.

Hvis der vindes på nummer 3, er gevinsten 3.5 gange indsatsen.

Hvis der vindes på nummer 4, er gevinsten 8 gange indsatsen.

Når opdateringen af kontoen er slut, meddeler BANKØREN SPILLERen, hvilken status hans konto har. Hvis SPILLERen sætter hele sin konto på spil og taber, giver BANKØREN ham valget mellem at indbetale et nyt beløb på kontoen eller forlade spillet. Når SPILLERen forlader spillet, takker BANKØREN ham for spillet og opfordrer ham til at komme igen

en anden gang.

Hvis SPILLERen overtræder en spilleregulering, modtager han en advarsel af BANKØRen. Hvis SPILLERen har modtaget 4 sådanne advarsler, bliver han af BANKØRen overgivet til UDSMIDERen. I dette tilfælde takker BANKØRen ikke SPILLERen.

slut BANKØR

en UDSMIDER

handlingsmønster: beder på opfordring af BANKØRen SPILLERen om at forlade lokalet uden at vække opsigt.

slut UDSMIDER

en KLOKKE

datastruktur: er udformet som en "ding-dong"-klokke. Står i forbindelse med en knap, som BANKØRen alene kan betjene.

handlingsmønster: ringer, hver gang BANKØRen betjener knappen.

slut KLOKKE

slut KASINO system.

Vi går derefter over til en mere formel beskrivelse af KASINOets komponenter. Lad os først se på CROUPIERen, som er en så simpel komponent, at vi kan beskrive den i én enkelt procedure:

proc croupier

gentag

betjen spiller

hvis spilleren vil spille, så betjen hjulet

indtil spilleren vil gå

slutproc croupier

SPILLER-komponenten er også ret simpel, og vi kan beskrive den i en enkelt procedure:

proc spiller

croupieren beder spilleren vælge et nummer

hvis spilleren ønsker dette så

skal han gøre sin indsats

ellers

skal bankøren udbetale ham, hvad der står på
kontoen, og

spillet standses

slut

slutproc spiller

Det bemærkes, at spilleren her er fremstillet omtrent som croupieren ser ham. Spillerens mulighed for at sætte penge i spillebanken er ikke med i denne beskrivelse. Grunden hertil fremgår af det følgende, som indeholder en beskrivelse af BANKØREN. Da denne er en meget fortagssom herre, vælger jeg at dele BANKØR-komponenten op i en række procedurer ("tasks") af hvilke den første netop skal administrere SPILLERens indskud:

proc indskud

gentag

bed spilleren om indskuddet;

kontrollér de penge, han betaler med

hvis pengene er falske så

afvis indskuddet;

advar spilleren

ellers

sæt indskuddet ind på spillerens konto

slut

indtil indskuddet er ok eller der er givet 4 advarsler

hvis der er givet 4 advarsler så tilkald udsmideren

slutproc indskud

Hver gang, der spilles, skal SPILLERen gøre en indsats (i hele kroner). Også dette tager BANKØREN sig af, og vi beskriver denne del af BANKØREns arbejde i følgende procedure:

proc indsats

gentag

bed spilleren om indsatsen

hvis indsatsen er større end kontoen så

spørg spilleren, om han vil indsætte flere penge

hvis svaret er ja så

bed ham indbetale pengene

ellers

bed ham gøre en mindre indsats;

advar ham mod forsøg på at overtrække kontoen

slut

slut

hvis spilleren forsøger at indsætte ørebeløb så

gør ham opmærksom på spillereglerne;

giv ham en advarsel

slut

indtil indsatsen er ok eller der er givet fire advarsler

hvis der er givet 4 advarsler så tilkald udsmidere

slutproc indsats

Efter hvert spil, skal resultatet gøres op. Denne del af BANKØRens arbejde beskrives i følgende procedure:

proc opgør

hvis spilleren har gættet rigtigt så

læg gevinsten til hans konto;

ring med klokken;

ønsk spilleren til lykke;

oplys ham om status for hans konto

ellers (spilleren har altså tabt)

træk indsatsen fra kontoen;

beklag spillerens uheld;

oplys ham om status for kontoen

hvis kontoen er blevet tom så

spørg spilleren om han vil indskyde et nyt beløb

hvis svaret er ja så

bed ham indbetale det

ellers

tag venligt afsked med ham

slut

slut

slut

slutproc opgør

Når SPILLERen forlader spillet - og stadig er i nåde - udbetaler BANKØREN ham den eventuelle konto og tager pænt afsked med ham:

proc afsked

hvis kontoen ikke er tom sd udbetal indestående slut
tak spilleren for spillet og bed ham komme snart igen
endproc afgang

Hvis SPILLERen er faldet i unåde hos BANKØREN på grund af sin mindre heldige opførsel, tager UDSMIDERen sig af SPILLERen:

proc udsvider

sig til spilleren, at han er uønsket i lokalet;
bed ham forlade dette uden at vække opsigt;
sig, at han kan få sine penge ved kassen ved udgangen;
stop spillet

endproc udsvider

HJUL-komponenten er i princippet meget simpel:

proc hjul

lad hjulet dreje;
aflæs resultatet, når det er standset;
foretag en opgørelse af resultatet

endproc hjul

Af alle komponenter er KLOKKE den simpleste:

proc klokke

ring hver gang bankøren giver signal
slutproc klokke

Ved programmeringen af simulatoren skal søge at få repræsenteret de tilstande og størrelser, der optræder i det virkelige system, på en sådan form, at de kan bruges i det programmeringssprog, vi har til rådighed. De forskellige handlinger og beslutninger, systemets komponenter kan udføre, skal også beskrives i dette sprog.

Vi må desuden tage hensyn til, at vi ikke mere har de fysiske begrænsninger, som det virkelige system giver anledning til. Således kan en spiller ikke i virkeligheden holde på et nummer, der ikke findes på hjulet, men operatøren ved terminalen kan udmærket ved en fejl eller for at provokere systemet indtaste et nummer, som ikke findes. Sådanne operatørfejl må vi også søge at imødegå i vort program.

Det benyttede sprog er COMAL.

Jeg vælger at begynde med at programmere SPILLER-komponenten. Programmet ser således ud:

```

0390 PROC SPILLER
0400   PRINT
0410   LET OK=TRUE
0420   REPEAT
0430     INPUT "VÆLG DERES NUMMER: ", GAET
0440     IF GAET=0 THEN
0450       LET UD=TRUE
0460       EXEC AFSKED
0470     ELSE
0480       LET OK=<GAET>0) AND (GAET<5) AND (GAET=INT(GAET))
0490       IF OK THEN
0500         EXEC INDSATS
0510       ELSE
0520         PRINT "OPERATØRFEJL! UMULIG SITUATION!"
0530         INPUT "ØNSKER DE INSTRUKTION ('JA' ELLER 'NEJ')? ", SVAR$
0540         IF SVAR$="JA" THEN EXEC INSTR
0550       ENDIF
0560     ENDIF
0570   UNTIL UD OR OK
0580 ENDPROC   SPILLER

```

I linje 410 tildeles den Boolske variabel OK værdien TRUE (ækvivalent med den numeriske konstant 1). Når sætningen i linje 430 udføres, skal operatøren indtaste

det nummer, spilleren holder på. Hvis spilleren ikke ønsker at spille mere, skal der indtastes et 0. Tallet 0 repræsenterer altså en beslutning hos spilleren om at forlade spillet. Hvis der indtastes et 0, er det Boolske udtryk i sætningen i linje 440 sandt, og sætningerne i linje 450 og 460 udføres. I sætningen i linje 450 får den Boolske variabel UD tildelt værdien TRUE, hvorpå proceduren AFSKED udføres. Hvis der ikke indtastes et 0, udføres sætningen i linje 480, hvorved den Boolske variabel OK tildeles værdien af det Boolske udtryk:

```
(GAET>0) AND (GAET<5) AND (GAET=INT(GAET))
```

OK får altså tildelt værdien TRUE, hvis det valgte nummer er legalt for spillet (dvs. er større end 0 og mindre end 5 og et helt tal). Hvis det er tilfældet, kan proceduren INDSATS udføres. I modsat fald foreligger der en operatørfejl, og OK får tildelt værdien FALSE (svarende til den numeriske konstant 0). Dette bevirker, at der udskrives en vejledning, og spilleren får tilbudt instruktion i spillet (520 - 540).

Proceduren afsluttes, når der er indtastet et legalt spillenummer - OK er sand - eller spilleren vil forlade spillet - UD er sand.

Dernæst vil jeg vise, hvorledes den del af BANKØR-komponenten, vi har kaldt INDSKUD, kan programmeres:

```
0640 PROC INDSKUD
0650 PRINT
0660 PRINT
0670 REPEAT
0680 LET OK=TRUE
0690 INPUT "HVOR STORT BELØB VIL DE INDSKYDE? ". INDBET
0700 CASE TRUE OF
0710 LET KONTO=KONTO+INDBET
0720 WHEN INDBET<0
0730 PRINT "DERES FALSKE PENGE KAN DE BEHOLDE FOR DEM SELV!"
0740 LET ADV=ADV+1; OK=FALSE
0750 WHEN INDBET=0
0760 PRINT "JEG TROEDE, DE MENTE DET ALVORLIGT!"
0770 LET ADV=ADV+1; OK=FALSE
0780 WHEN INDBET<1
0790 PRINT "DET DER ØREPJAT GIVER VI IKKE! MENNESKEPENGE OK=0"
0800 LET OK=FALSE; ADV=ADV+1
```

```

0810     WHEN INDBET<>INT<INDBET>
0820     PRINT "DRIKKEPENGE! MANGE TAK"
0830     LET INDBET=INT<INDBET>
0840     LET KONTO=KONTO+INDBET
0850     ENDCASE
0860     UNTIL OK OR <ADV=4>
0870     IF ADV=4 THEN EXEC UDSMIDER
0880     ENDPROC     INDSKUD

```

Heri anvendes atter den Booleske variabel OK, der sættes til TRUE i linje 680. Derpå spørger programmet (altså BANKØRen), hvor stort et beløb, spilleren vil indsætte (690), og i afsnittet fra linje 700 til linje 850 bliver det indtastede beløb undersøgt. Jeg har her valgt at undersøge forskellige fejlmuligheder, idet jeg lader disse forestille eller repræsentere forskellige tildragelser, som faktisk kunne tænkes at forekomme i det virkelige spil. Her har man naturligvis nogenlunde frit valg i forhold til den primære beskrivelse, og man kan lade en ren underholdningseffekt afgøre, hvilke reaktioner, man vil lade simulatoren demonstrere. Det er således programmørens valg at lade indtastning af et negativt beløb fortolke som et forsøg på at betale med falske penge. Forsøg af den art, må naturligvis føre til en advarsel, hvilket netop repræsenteres af tildelingen i linje 740:

```
ADV=ADV+1
```

Samtidig bliver den Booleske variabel OK sat til FALSE, som tegn på, at der foreligger en illegal tilstand. Proceduren afsluttes, hvis der indtastes en legal sum penge, eller hvis der er givet 4 advarsler. I sidste tilfælde bliver spilleren smidt ud af kasinoet på grund af sin umodne adfærd.

Af hensyn til de læsere, som ikke kender CASE-strukturen (multi-forgreningen), skal det bemærkes, at sætningen i linje 710 udgør det såkaldte alternative tilfælde, som udføres, hvis ingen af de angivne udsagn (efter WHEN'erne) er sande. I dette eksempel er det alternative tilfælde

særlig vigtigt, idet det repræsenterer den situation, at spilleren har opført sig korrekt, mens de øvrige tilfælde alle behandler mere eller mindre utilbørlig optræden hos spilleren.

Som eksempel på programmering af mere "teknisk betonedede" procedurer, vises dernæst et forslag til programmering af HJUL-komponenten:

```

1340 PROC HJUL
1350 LET TAL=INT(RND(0)*9)+1
1360 CASE TAL OF
1370 LET UDFALD=4; FAK=8
1380 WHEN 1, 2, 3
1390 LET UDFALD=1; FAK=2
1400 WHEN 4, 5, 6
1410 LET UDFALD=2; FAK=2
1420 WHEN 7, 8
1430 LET UDFALD=3; FAK=3.5
1440 ENDCASE
1450 PRINT
1460 PRINT
1470 PRINT "DET BLEV NUMMER"; UDFALD; "DER KOM UD. "
1480 EXEC OPGØR
1490 ENDPROC HJUL

```

Man kan her bemærke, at sandsynlighedsfordelingen for spillet frembringes ved en transformation af en jævn belægning på udfaldsrummet: $\{1, 2, 3, 4, 5, 6, 7, 8, 9\}$. Endvidere ses det, at den stokastiske variabel FAK ("gevinst-faktoren") får tildelt sin værdi sammen med udfaldet af spillet.

Til slut vises, hvorledes indledningen af programmet er organiseret. Læseren bedes bemærke, at de Booleske konstanter TRUE og FALSE defineres i dette afsnit. Proceduren, som udfører CROUPIER-komponenten er også vist. Læg mærke til, at HJUL-komponenten kun udføres, hvis spilleren ønsker at fortsætte, altså hvis UD har værdien FALSE. Hvis UD har værdien TRUE, bringes proceduren CROUPIER til afslutning af sætningen

UNTIL UD


sætningen i linje 240 udføres som den næste. Da der er tale om en END-sætning, standser simulatoren.

```

0080 REM (*INITIALISERING*)
0090 REM
0100 RANDOMIZE
0110 LET TRUE=1; FALSE=0
0120 LET UD=FALSE
0130 LET ADV=0; SADV=0
0140 DIM SVAR$(5)
0150 REM (*ENDINIT*)
0160 REM
0170 REM //-----//
0180 REM
0190 REM (*HOVEDPROGRAM*)
0200 INPUT "ØNSKER DE INSTRUKTION OM SPILLET (JA ELLER NEJ) ", SVAR$
0210 IF SVAR$="JA" THEN EXEC INSTR
0220 EXEC INDSKUD
0230 IF OK THEN EXEC CROUPIER
0240 END (*SLUT HOVEDPROGRAM*)
0250 REM
0260 REM //-----//
0270 REM (*PROCEDURER*)
0280 REM //-----//
0290 REM
0300 PROC CROUPIER
0310 REPEAT
0320 EXEC SPILLER
0330 IF UD=FALSE THEN EXEC HJUL
0340 UNTIL UD
0350 ENDPROC CROUPIER

```

Hele programmet er organiseret således:



```

0010 REM (*HER BEGYNDER SIMULATOREN 'LILLE KASINO'*)
0020 REM (*SKREVET FOR DATO, TØNDER*)
0030 REM (*AF BØRGE R. CHRISTENSEN*)
0040 REM (*DENNE VERSION ER AF 770422*)
0050 REM
0060 REM //-----//
0070 REM
0080 REM (*INITIALISERING*)
0090 REM
0100 RANDOMIZE
0110 LET TRUE=1; FALSE=0
0120 LET UD=FALSE
0130 LET ADV=0; SADV=0
0140 DIM SVAR$(5)
0150 REM (*ENDINIT*)
0160 REM
0170 REM //-----//
0180 REM
0190 REM (*HOVEDPROGRAM*)
0200 INPUT "ØNSKER DE INSTRUKTION OM SPILLET (JA ELLER NEJ) ", SVAR$
0210 IF SVAR$="JA" THEN EXEC INSTR
0220 EXEC INDSKUD
0230 IF OK THEN EXEC CROUPIER
0240 END (*SLUT HOVEDPROGRAM*)
0250 REM
0260 REM //-----//
0270 REM (*PROCEDURER*)
0280 REM //-----//
0290 REM
0300 PROC CROUPIER
0310 REPEAT
0320 EXEC SPILLER
0330 IF UD=FALSE THEN EXEC HJUL
0340 UNTIL UD
0350 ENDPROC CROUPIER
0360 REM
0370 REM //-----//
0380 REM
0390 PROC SPILLER
0400 PRINT
0410 LET OK=TRUE
0420 REPEAT
0430 INPUT "VÆLG DERES NUMMER: ", GAET
0440 IF GAET=0 THEN
0450 LET UD=TRUE
0460 EXEC AFSKED
0470 ELSE
0480 LET OK=(GAET>0) AND (GAET<5) AND (GAET=INT(GAET))
0490 IF OK THEN
0500 EXEC INDSATS
0510 ELSE
0520 PRINT "OPERATØRFEJL! UMULIG SITUATION!"
0530 INPUT "ØNSKER DE INSTRUKTION ('JA' ELLER 'NEJ')? ", SVAR$
0540 IF SVAR$="JA" THEN EXEC INSTR
0550 ENDF
0560 ENDF
0570 UNTIL UD OR OK
0580 ENDPROC SPILLER
0590 REM
0600 REM //-----//

```

```

0610 REM
0620 PROC INDSKUD
0630 PRINT
0640 PRINT
0650 REPEAT
0660 LET OK=TRUE
0670 INPUT "HVOR STORT BELØB VIL DE INDSKYDE? ", INDBET
0680 CASE TRUE OF
0690 LET KONTO=KONTO+INDBET
0700 WHEN INDBET<0
0710 PRINT "DERES FALSKE PENGE KAN DE BEHOLDE FOR DEM SELV!"
0720 LET ADV=ADV+1; OK=FALSE
0730 WHEN INDBET=0
0740 PRINT "JEG TROEDE, DE MENTE DET ALVORLIGT!"
0750 LET ADV=ADV+1; OK=FALSE
0760 WHEN INDBET<1
0770 PRINT "DET DER ØREPJAT GIVER VI IKKE! MENNESKEPENGE OM VI MAA BE!"
0780 LET OK=FALSE; ADV=ADV+1
0790 WHEN INDBET<>INT<INDBET>
0800 PRINT "DRIKKEPENGE! MANGE TAK"
0810 LET INDBET=INT<INDBET>
0820 LET KONTO=KONTO+INDBET
0830 ENDCASE
0840 UNTIL OK OR <ADV=4>
0850 IF ADV=4 THEN EXEC UDSMIDER
0860 ENDPROC INDSKUD
0870 REM
0880 REM //-----//
0890 REM
0900 PROC AFSKED
0910 PRINT
0920 PRINT
0930 IF KONTO<>0 THEN
0940 PRINT "DE FAAR UDBETALT DERES INDESTAENDE, STORT KR. "; KONTO
0950 ENDF
0960 PRINT "TAK FOR SPILLET. "
0970 IF ADV<2 THEN PRINT "DET HAR VAERET OS EN FORNØJELSE. "
0980 PRINT "KOM SNART IGEN. "
0990 ENDPROC AFSKED
1000 REM
1010 REM //-----//
1020 REM
1030 PROC OPGØR
1040 PRINT
1050 PRINT
1060 IF GAET=UDFALD THEN
1070 LET KONTO=KONTO+INDSATS*FAK
1080 EXEC KLOKKE
1090 PRINT "TILLYKKE!"
1100 PRINT "DE HAR VUNDET"; INDSATS*FAK; "KR. ", OG DE HAR NU
1110 PRINT KONTO; "KR. PAA DERES KONTO. "
1120 ELSE
1130 LET KONTO=KONTO-INDSATS
1140 PRINT "DE HAR DESVAERRE TABT DERES INDSATS. STOR KR. "; INDSATS
1150 PRINT "BEDRE HELD NÆSTE GANG. "
1160 PRINT "DE HAR NU"; KONTO; "KR. PAA DERES KONTO"
1170 IF KONTO=0 THEN
1180 PRINT "VIL DE INDSKYDE ET NYT BELØB";
1190 INPUT SVAR$

```

```

1200     IF SVAR$="JA" THEN
1210         EXEC INDSKUD
1220     ELSE
1230         EXEC AFSKED
1240         LET UD=TRUE
1250     ENDIF
1260 ENDIF
1270 ENDIF
1280 ENDPROC OPGØR
1290 REM
1300 REM //-----//
1310 REM
1320 PROC HJUL
1330     LET TAL=INT(RND(0)*9)+1
1340     CASE TAL OF
1350         LET UDFALD=4; FAK=8
1360     WHEN 1, 2, 3
1370         LET UDFALD=1; FAK=2
1380     WHEN 4, 5, 6
1390         LET UDFALD=2; FAK=2
1400     WHEN 7, 8
1410         LET UDFALD=3; FAK=3.5
1420     ENDCASE
1430     PRINT
1440     PRINT
1450     PRINT "DET BLEV NUMMER"; UDFALD; "DER KOM UD. "
1460     EXEC OPGØR
1470 ENDPROC HJUL
1480 REM
1490 REM //-----//
1500 REM
1510 PROC KLOKKE
1520     FOR I=1 TO 2*FAK
1530         PRINT "<?>";
1540     NEXT I
1550 ENDPROC KLOKKE
1560 REM
1570 REM //-----//
1580 REM
1590 PROC INDSATS
1600     PRINT
1610     PRINT
1620     REPEAT
1630         LET OK=TRUE
1640         INPUT "DERES INDSATS? ", INDSATS
1650         CASE TRUE OF
1660             REM *(INDSATS I ORDEN*)
1670         WHEN INDSATS>KONTO
1680             IF SADV=1 THEN EXEC UDSMIDER
1690             PRINT "SAA MANGE PENGE HAR DE IKKE"
1700             INPUT "ØNSKER DE AT INDSKYDE FLERE I BANKEN ", SVAR$
1710             IF SVAR$="JA" THEN
1720                 EXEC INDSKUD
1730             ELSE
1740                 PRINT "SAA MAA DE GØRE EN MINDRE INDSATS. "
1750                 PRINT "DE HAR KUN"; KONTO; "KR. TI DERES RÅDGIVER"
1760                 PRINT "JEG MAA ADVARE DEM PÅ DET ØRETTLIGE SÅDANLIGT"
1770                 PRINT "FORLØBE OG OVERBÅN"
1780                 LET ADV=ADV+1. SADV=1; OK=1
1790             ENDIF

```

```

1800     WHEN INDSATS<0
1810         PRINT "SPILLEMAERKER FRA AALBORG MODTAGES IKKE. "
1820         LET ADV=ADV+1; OK=FALSE
1830     WHEN INDSATS<>INT<INDSATS>
1840         PRINT "KUN HELE KRONER. FEDTSYL!"
1850         LET OK=FALSE
1860     WHEN INDSATS=0
1870         PRINT "VAER VENLIG AT TAGE SPILLET ALVORLIGT. "
1880         LET ADV=ADV+1; OK=FALSE
1890     ENDCASE
1900     UNTIL OK OR <ADV=4>
1910     IF ADV=4 THEN EXEC UDSMIDER
1920 ENDPROC INDSATS
1930 REM
1940 REM //-----//
1950 REM
1960 PROC INSTR
1970     PRINT
1980     PRINT "NAAR CROUPIEREN OPFORDERER DEM TIL AT VAELGE ET NUMMER. "
1990     PRINT "HAR DE FØLGENDE MULIGHEDER, HVIS DE ØNSKER AT SPILLE:"
2000     PRINT "1. 2. 3 ELLER 4. "
2010     PRINT "HVIS DE ØNSKER AT FORLADE SPILLET, SKAL DE VAELGE NUMMERET 0"
2020     PRINT "ANDRE NUMRE ER IKKE TILLADT. "
2030     PRINT "DERPAA BEDER BANKØREN OM DERES INDSATS, SOM SKAL"
2040     PRINT "VAERE I HELE KRONER. "
2050     PRINT
2060     PRINT "HVIS DERES NUMMER IKKE KOMMER UD, HAR DE TABT INDSATSEN. "
2070     PRINT
2080     PRINT "HVIS DE VINDER PAA 1 ELLER 2, "
2090     PRINT "VINDER DE DERES INDSATS 2 GANGE. "
2100     PRINT "HVIS DE VINDER PAA NUMMER 3, VINDER DE DERES INDSATS 3.5 GANGE"
2110     PRINT "HVIS DE VINDER PAA NUMMER 4, VINDER DE DERES INDSATS 8 GANGE"
2120     PRINT
2130     PRINT "HVIS DE IKKE TAGER SPILLET ALVORLIGT, ELLER HVIS DE"
2140     PRINT "BEGAAR GROVE FEJL, FAAR DE EN ADVARSEL AF BANKØREN. "
2150     PRINT "HVIS DE FAAR 4 ADVARSLER AF BANKØREN, BLIVER DE"
2160     PRINT "BORTVIST FRA SPILLET. "
2170     PRINT
2180     PRINT "NAAR DE HAR LAEST INSTRUKTIONEN, SAA TAST 'CON'"
2190     STOP
2200 ENDPROC INSTR
2210 REM
2220 REM //-----//
2230 REM
2240 PROC UDSMIDER
2250     PRINT "DERES TILSTEDEVAERELSE I KASINOETS LOKALER ER UØNSKET. "
2260     PRINT "FORLAD VAENLIGST ETABLISSEMENTET UDEN AT VAEKKE OPSIGT. "
2270     STOP
2280 ENDPROC UDSMIDER
2290 REM
2300 REM (*HER ENDER SIMULATOREN 'LILLE KASINO'*)
2310 REM
2320 REM //-----//

```